
ABSTRACT: The Macaulay2 package Cremona.m2 performs some computa-
tions on rational and birational maps between irreducible projective varieties.
For instance, it provides methods to compute degrees and projective degrees of
rational maps without any theoretical limitation, from which is derived a general
method to compute the push-forward to projective space of Segre classes. More-
over, the computations can be done both deterministically and probabilistically.
We give here a brief description of the methods and algorithms implemented.

INTRODUCTION. In this note we describe the computational package Cremona.m2,
included with [Macaulay2] since version 1.9. A first rudimentary version of this
package has been already used in an essential way in [Staglianò 2016] (it was
originally named bir.m2), and recent applications can be found in [Staglianò 2018;
Russo and Staglianò 2017]. Here we describe version 4.2.2 of the package.

Cremona.m2 performs computations on rational and birational maps between
absolutely irreducible projective varieties over a field K. Among other things, it
provides general methods to compute projective degrees of rational maps, from
which, as is well known (see Proposition 1.2), one can interpret them as methods
to compute the push-forward to projective space of Segre classes. The algorithms
are naively derived from the mathematical definitions, with the advantages of being
obvious, quite general and easily implemented. Moreover, all the methods (where
this may make sense) are available both in a probabilistic version and in a deter-
ministic version, and one can switch from one to the other with a boolean option
named MathMode.

In Section 1, we will describe the main methods provided by the package and the
algorithms implemented. Most of these have already been described in [Staglianò
2016, Section 2], but here we will consider a more general setting. For instance,
Algorithm 1.3 for computing homogeneous components of kernels of homoge-
neous ring maps was presented in [Staglianò 2016, Algorithm 2.5] requiring that

the map was between polynomial rings. In Section 2, we will show how these
methods work in some particular examples, concluding with an experimental com-
parison of the running times of one of these methods with the corresponding ones
proposed in [Helmer 2016] and [Harris 2017] (see also [Jost 2015]). For further
technical details we refer to the documentation of the package, which can be shown
using the command viewHelp Cremona.

We mention that the package RationalMaps.m2, by K. Schwede, D. Smolkin,
S. H. Hassanzadeh, and C. J. Bott, is another package included with Macaulay2
for working with rational maps. It mainly focuses on providing a general method
for inverting birational maps, which in some cases turns out to be competitive with
the corresponding method of Cremona.m2.

1. DESCRIPTION OF THE MAIN METHODS. Throughout, we shall use the follow-
ing notation. Let K denote a field; in practice, it can be for instance Q, a finite field,
or a fraction field of a polynomial ring over these. Let φ : X 99K Y be a rational
map from a subvariety X = V (I ) ⊆ Pn

= Proj(K[x0, . . . , xn]) to a subvariety
Y = V (J ) ⊆ Pm

= Proj(K[y0, . . . , ym]), which can be represented, although not
uniquely, by a homogeneous ring map

ϕ : K[y0, . . . , ym]/J → K[x0, . . . , xn]/I

of quotients of polynomial rings by homogeneous ideals. Sometimes we will de-
note by F0, . . . , Fm ∈K[x0, . . . , xn] homogeneous forms of the same degree such
that Fi := Fi + I = ϕ(yi ), for i = 0, . . . ,m. The common degree of these elements
will be denoted by δ.

From algebraic geometry to computational algebra. For each homogeneous ideal
a ⊆ K[x0, . . . , xn]/I (resp. b ⊆ K[y0, . . . , ym]/J ), we have a closed subscheme
V (a)⊆ X (resp. V (b)⊆ Y ), and the following basic formulae hold:1

φ(V (a))=V (ϕ−1(a)) and φ−1(V (b))=V ((ϕ(b)):(ϕ(y0), . . . ,ϕ(ym))
∞). (1-1)

In particular, the (closure of the) image of φ is defined by the kernel of ϕ. Several
issues concerning rational maps lead naturally to an examination of the left-hand
sides of (1-1), and the right-hand sides of (1-1) can be determined using Gröbner
basis techniques, whenever a and b are explicitly given. Furthermore, Macaulay2
provides useful commands such as preimage, kernel and saturate, so that the
required user skill level is quite low. The aim of the package Cremona.m2 is to
provide further tools.

1By abuse of notation, we consider φ as a morphism defined on the open set X \ V (F0, . . . , Fm).

https://github.com/Macaulay2/M2/blob/master/M2/Macaulay2/packages/RationalMaps.m2

Computing projective degrees. The projective degrees are the most basic invari-
ants of a rational map. Many others can be derived from them, such as, for instance,
the dimension and the degree of the base locus. For more details on the subject,
see [Harris 1992, Example 19.4, p. 240].

Definition 1.1 (projective degrees, [Harris 1992]). (1) The projective degrees
d0(φ), d1(φ), . . . , ddim X (φ) of the map φ are defined as the components of
the multidegree of the closure of the graph 0φ ⊂ Pn

×Pm .

(2) Equivalently, the i-th projective degree di (φ) can be defined in terms of dimen-
sion and degree of the closure of φ−1(L), where L is a general (m− dim X +i)-
dimensional linear subspace of Pm ; more precisely, di (φ)= degφ−1(L) if
dim φ−1(L)= i , and di (φ)= 0 otherwise.

In common computer algebra systems such as Macaulay2, it is easy to translate
Definition 1.1 into code. We now describe in more detail how this can be done.
All of this is implemented in the method projectiveDegrees; see Example 2.2
for an example using it.

Deterministic approach. Taking into account Definition 1.1(1), a bihomogeneous
ideal for 0φ in K[x0, . . . , xn, y0, . . . , ym] can be, for instance, obtained as

(I + ({yi F j − y j Fi , 0≤ i, j ≤ m})) : (F0, . . . , Fm)
∞. (1-2)

Therefore its multidegree can be computed in Macaulay2 with multidegree, which
implements an algorithm according to [Miller and Sturmfels 2005, p. 165].

Probabilistic approach. (See also [Staglianò 2016, Remark 2.4].) Taking into ac-
count Definition 1.1(2), if L is defined by an ideal IL , the second formula of (1-1)
tells us that φ−1(L) is defined by the saturation of the ideal (ϕ(IL)) by (F0, . . . , Fm)

in the ring K[x0, . . . , xn]/I. So replacing the word general with random in the
definition, we get a probabilistic algorithm that computes all the projective degrees.
Moreover, we can considerably speed up this algorithm by taking into account two
remarks: firstly, the saturation

ϕ(IL) : (F0, . . . , Fm)
∞

is the same as
ϕ(IL) : (λ0 F0+ · · ·+ λm Fm)

∞
,

where λ0, . . . , λm ∈K are general scalars; secondly, the i-th projective degree of
φ coincides with the (i−1)-th projective degree of the restriction of φ to a general
hyperplane section of X.

An alternative deterministic approach. Replacing the word general with symbolic
in Definition 1.1(2) gives us a deterministic algorithm for computing projective
degrees. For instance, in the case in which φ : Pn 99K Pn is a dominant rational

map, extending K to the fractional field of a polynomial ring K[a0, . . . , an], we
have that d0(φ) is the degree of the fiber of φ at the symbolic point [a0, . . . , an].

Some applications using projective degrees.
The degree of a rational map. The degree of the map φ : X 99K Y is the number of
isolated points in the inverse image of a general point of φ(X) over the algebraic
closure of K. This is the same as the ratio of d0(φ) and degφ(X), and thus it can
be explicitly computed. Let us note, however, that in several cases we do not need
to compute the kernel of ϕ. For instance, if X is a projective space, we are able to
pick an abundance of rational points of ϕ(X) and then we apply the second formula
of (1-1). Another special case is when d0(φ) is a prime number: here we have only
to establish if the image of φ is a linear subspace (e.g., applying Algorithm 1.3
with d = 1). The method provided by Cremona.m2 for this computation is named
degreeOfRationalMap.2

Methods related to this are isBirational and isDominant, with obvious
meaning. The latter does not compute the kernel of ϕ, but it uses an algorithm
that looks for dr (φ), where r = dim X − dim Y. More precisely, the algorithm is
based on the following fact: let Z ⊂ Y be a random 0-dimensional linear section
of Y ; if dim φ−1(Z)= dim X − dim Y ≥ 0, then φ is certainly dominant, otherwise
it is probably not dominant (see [Mumford 1988, Chapter I, § 8] or [Hartshorne
1977, Chapter II, Exercise 3.22]). When this last case occurs, it is generally easy
to find a nonzero element in the kernel of ϕ, and so this method turns out to be
very effective even in its deterministic version (see Example 2.1).

The Segre class. It is well known that one can deduce an algorithm computing the
push-forward to projective space of Segre classes from an algorithm computing
projective degrees of rational maps between projective varieties and vice versa.
Indeed, with our notation, we have the following:

Proposition 1.2 ([Fulton 1984, Proposition 4.4]; see also [Dolgachev 2011, Section
2.3; Aluffi 2003, Section 3]). Let B⊂ X be the subscheme defined by F0, . . . , Fm

and let ν : X ↪→ Pn be the inclusion. If H denotes the hyperplane class of Pn and
r = dim X, then the push-forward ν∗(s(B, X)) of the Segre class of B in X is

ν∗(s(B, X))=
dimB∑
k=0

((−1)r−k−1
r−k∑
i=0

(−1)i
(

r − k
i

)
δr−k−i dr−i (ϕ)) H n−k . (1-3)

The general method SegreClass, provided by Cremona.m2 for computing the
push-forward to projective spaces of Segre classes, does basically nothing more

2Notice that, in general, if the result of the probabilistic algorithm for degreeOfRationalMap is
wrong, it can be either too small or too large. However, as a consequence of [Hartshorne 1977, Chap-
ter III, Exercise 10.9], it should always provide a lower bound when the map is dominant between
smooth varieties.

than apply (1-3); see Example 2.3 for an example using this method. Furthermore,
applying one of the main results in [Aluffi 2003], a method is derived to com-
pute the push-forward to projective space of the Chern–Schwartz–MacPherson
class CSM(W ) of the support of a projective scheme W ; recall that the compo-
nent of dimension 0 of CSM(W ) is the topological Euler characteristic of the
support of W.

Computing homogeneous components of kernels. To compute, using Macaulay2,
the homogeneous component of degree d of the kernel of phi (= ϕ), one can
perform the command ideal image basis(d,kernel phi). This is equivalent
to the command kernel(phi,d) provided by Cremona.m2, but the latter uses the
following obvious algorithm.

Algorithm 1.3. Input: the ring map ϕ and an integer d .

Output: homogeneous component of degree d of the kernel of ϕ.

• Find vector space bases G0, . . . ,Gr of (K[y0, . . . , ym]/J )d and H0, . . . , Hs

of Id δ, where subscripts stand for homogeneous components.

• Take generic linear combinations G=
∑r

i=0 ai Gi and H=
∑s

j=0 b j H j , and
find a basis of solutions for the homogeneous linear system obtained by re-
quiring that the polynomial

G(F0, . . . , Fm)−H ∈ K[a0, . . . , ar , b0, . . . , bs][x0, . . . , xn]

vanishes identically.

• For each vector (â0, . . . , âr , b̂0, . . . , b̂s) ∈ Kr+s+2 obtained in the previous
step, replace in G the coefficients a0, . . . , ar with â0, . . . , âr ; return all these
elements.

For small values of d, applying Algorithm 1.3 may turn out to be much faster
than computing a list of generators of the kernel of the map; see for instance
Example 2.1 below.

Inverting birational maps. General algorithms for inverting birational maps are
known. One of them is implemented in the package Parametrization.m2 by
J. Boehm, and the method inverseMap of Cremona.m2 uses the same one for
the general case as well. However, when the source X of the rational map φ is
a projective space and a further technical condition is satisfied, then it uses the
following powerful algorithm.

Algorithm 1.4 ([Russo and Simis 2001]; see also [Simis 2004]). Input: the
ring map ϕ (assuming that φ is birational and further conditions are satisfied).

Output: a ring map representing the inverse map of φ.

https://github.com/Macaulay2/M2/blob/master/M2/Macaulay2/packages/Parametrization.m2

• Find generators {(L0, j , . . . , Lm, j )} j=1,...,q for the module of linear syzygies
of (F0, . . . , Fm).

• Compute the Jacobian matrix 2 of the bihomogeneous forms{ m∑
i=0

yi L i, j

}
j=1,...,q

with respect to the variables x0, . . . , xn and consider the map of graded free
modules (2 mod J ) : (k[y0, . . . , ym]/J )n+1

→ (k[y0, . . . , ym]/J )q .

• Return the map defined by a generator G = (G0, . . . ,Gn) for the kernel of
(2 mod J ).

Remark 1.5. One of the main features of the package RationalMaps.m2, by
Schwede, Smolkin, Hassanzadeh, and Bott, is a method for inverting birational
maps, which, in the case when Algorithm 1.4 does not apply, appears to be quite
competitive with the method inverseMap of Cremona.m2.

Heuristic approach. The method approximateInverseMap provides a heuristic
approach to compute the inverse of a birational map modulo a change of coordinate.
The idea of the algorithm is to try to construct the base locus of the inverse by
looking for the images of general linear sections. Consider, for simplicity, the case
in which φ : Pn 99K Pn ′ is a Cremona transformation. Then, by taking the images
of n+ 1 general hyperplanes in Pn, we form a linear system of hypersurfaces in
Pn ′ of degree d1(φ) which defines a rational map ψ : Pn ′ 99K Pn such that ψ ◦φ is
a (linear) isomorphism; i.e., we find an approximation of φ−1. Next, we can fix the
error of the approximation by observing that we have φ−1

= (ψ ◦φ)−1
◦ψ . It is sur-

prising that this method turns to be effective in examples where other deterministic
algorithms seem to run endlessly; see for instance Example 2.1 below.

2. EXAMPLES. In this section, we show how the methods described in Section 1
can be applied in some particular examples. We note that the package Cremona.m2
provides the data type RationalMap, but here we will use the more familiar type
RingMap. For brevity, we will omit irrelevant output lines. We start with an ex-
ample reviewing the construction given in [Staglianò 2016] of a quadro-quadric
Cremona transformation of P20.

Example 2.1. The code below constructs a ring map psi representing a rational
map ψ : P16 99K P20. Precisely, the algorithm for constructing ψ is as follows:
take E ⊂ P7 to be a 3-dimensional edge variety of degree 7, namely, the residual
intersection of P1

×P3
⊂ P7 with a general quadric in P7 containing one of the

P3’s of the rulings of P1
×P3

⊂ P7; next, see E ⊂ P7 embedded in a hyperplane
of P8 and take the birational map φ : P8 99K P16 defined by the quadrics of P8

containing E ; take ψ : P16 99K P20 to be the map defined by the quadrics of P16

https://github.com/Macaulay2/M2/blob/master/M2/Macaulay2/packages/RationalMaps.m2

containing the image of φ. For the first part of this construction, we use the package
Cremona.m2 only to shorten the code.
Macaulay2, version 1.11
with packages: ConwayPolynomials, Elimination, IntegralClosure, InverseSystems,

LLLBases, PrimaryDecomposition, ReesAlgebra, TangentCone
i1 : loadPackage "Cremona";
i2 : K = ZZ/70001;
i3 : P8 = K[t_0..t_8]; E = saturate(minors(2,genericMatrix(P8,4,2))+sum(

(ideal(t_0..t_7)*ideal(t_0..t_3))_*,u->random(K)*u),ideal(t_0..t_3)) + t_8;
i5 : psi = toMap kernel(toMap(E,2),2);

Up to this point, the computation was standard. But now we want to determine
the homogeneous ideal of Z := ψ(P16)⊂ P20, which turns out to be generated by
quadrics. Computing this using kernel psi seems an impossible task, but it is
elementary using kernel(psi,2). So we can consider ψ as a dominant rational
map ψ : P16 99K Z ⊂ P20.
i6 : time Z = kernel(psi,2);

-- used 2.84998 seconds
i7 : psi = toMap(psi,Dominant=>Z);

The map ψ turns out to be not only dominant but birational.
i8 : time degreeOfRationalMap psi

-- used 2.11216 seconds
o8 = 1

We now want to compute the inverse of ψ . This is a case where inverseMap
can apply Algorithm 1.4, but the running time is several hours. We can perform
this computation in seconds by using approximateInverseMap.
i9 : time psi’ = approximateInverseMap(psi,CodimBsInv=>10,MathMode=>true);

-- used 15.9724 seconds

A Cremona transformation ω of P20 is then obtained combining ψ−1 and Z as
follows.
i10 : omega = toMap(lift(matrix psi’,ring Z)|gens Z);

Even checking just the dominance of ω, by computing kernel omega, seems
an impossible task, but it can be done quickly with isDominant.
i11 : time isDominant(omega,MathMode=>true)

-- used 0.100369 seconds
o11 = true

We now check that our map is birational and find its inverse using Algorithm 1.4.
i12 : time isBirational omega

-- used 0.0366468 seconds
o12 = true
i13 : time inverseMap omega;

-- used 0.0717518 seconds

Example 2.2. Here, we use the probabilistic versions of some methods. Take M
to be a generic 3× 5 matrix of linear forms on P6, and, let φ : P6 99K G(2, 4)⊂ P9

be the rational map defined by the 3× 3 minors of M (its base locus is a smooth
threefold scroll over a plane).

i14 : P6 = K[x_0..x_6]; M = matrix pack(5,for i from 1 to 15 list random(1,P6));
i16 : phi = toMap(minors(3,M),Dominant=>2);

We check that the map is birational and compute its inverse.
i17 : time isBirational phi

-- used 0.217607 seconds
o17 = true
i18 : time psi = inverseMap phi;

-- used 1.39511 seconds

Now we compute the multidegrees of φ and φ−1.
i19 : time (projectiveDegrees phi, projectiveDegrees psi)

-- used 1.37582 seconds
o19 = ({1, 3, 9, 17, 21, 15, 5}, {5, 15, 21, 17, 9, 3, 1})

We also compute the push-forward to P6 (resp. P9) of the Segre class of the
base locus of φ (resp. φ−1) in P6 (resp. in G(2, 4)). As usual, H denotes the
hyperplane class.
i20 : time (SegreClass phi, SegreClass psi)

-- used 1.43359 seconds
6 5 4 3 9 8 7 6 5

o20 = (- 680H + 228H - 60H + 10H , 728H - 588H + 276H - 98H + 24H )

Example 2.3. In this example, we use the deterministic version of the method
SegreClass. We take Y ⊂ P11 to be the dual quartic hypersurface of

P1
× Q4

⊂ P11∗,

where Q4
⊂P5 is a smooth quadric hypersurface, and take X ⊂ Y to be the singular

locus of Y. We then compute the push-forward to the Chow ring of P11 of the Segre
class both of X in Y and of X in P11 working over the Galois field GF(3312).
i21 : P11 = GF(331^2)[x_0..x_11];
i22 : Y = ideal sum(first entries gens minors(2,genericMatrix(P11,6,2)),t->t^2);
i23 : X = sub(ideal jacobian Y,P11/Y);
i24 : time SegreClass(X, MathMode=>true) -- push-forward of s(X,Y)

-- used 0.789986 seconds
11 10 9 8 7 6 5 4 3

o24 = 507384H - 137052H + 35532H - 9018H + 2340H - 658H + 204H - 64H + 16H
i25 : time SegreClass(lift(X,P11), MathMode=>true) -- push-forward of s(X,P^11)

-- used 0.846234 seconds
11 10 9 8 7 6 5 4 3

o25 = 313568H - 101712H + 30636H - 8866H + 2532H - 720H + 198H - 48H + 8H

Example 2.4. Here we experimentally measure the probability of obtaining an in-
correct answer using the probabilistic version of the method projectiveDegrees
with a simple example of a birational map φ : G(1, 3) 99K P4 defined over K. We
define a procedure which computes this probability as a function of the field K.



Staglianò :::: A Macaulay2 package for computations with rational maps 69

field Q Z/70001 GF(38) Z/101 Z/31
probability 0.0 0.0 0.002 0.074 0.253

Table 1. Incorrect outputs of a probabilistic method.

In Table 1, we report the results obtained by running the procedure with various
fields.
i26 : p = (K) -> (

x := local x; R := K[x_0..x_4];
phi := inverseMap toMap(minors(2,matrix{{x_0..x_3},{x_1..x_4}}),Dominant=>2);
m := projectiveDegrees(phi,MathMode=>true);
0.1 * # select(1000,i -> projectiveDegrees phi != m));

Example 2.5. Lastly, we deal with an experimental comparison of the method
SegreClass of Cremona.m2 and the corresponding ones of other Macaulay2 pack-
ages. Precisely, we want to compare the method SegreClass against the corre-
sponding methods of the packages CharacteristicClasses.m2 version 2.0, by
M. Helmer and C. Jost (see [Helmer 2016; Jost 2015]), which provides a probabilis-
tic method; and FMPIntersectionTheory.m2 version 0.1, by C. Harris [2017],
which provides a deterministic method. Since the former puts restrictions on the
ambient variety, we will only consider examples where the ambient is a projective
space. We are unable to determine precisely which is the fastest among all the
methods and which, in the probabilistic case, has highest probability of giving the
correct answer. We just summarize in Table 2 the running times for some special
examples. Below is the code from which we obtained the first row of the table.
i27 : loadPackage "CharacteristicClasses"; loadPackage "FMPIntersectionTheory";
i29 : X = last(P5=ZZ/16411[vars(0..5)],ideal(random(3,P5),random(3,P5),random(4,P5)));
i30 : (time Segre X,time SegreClass X,time segreClass X,time SegreClass(X,MathMode=>true));

-- used 0.1511 seconds
-- used 1.00936 seconds
-- used 34.1471 seconds
-- used 74.572 seconds

input CC
Cremona
(prob.) FMPIntTh

Cremona
(det.)

complete int. of type (3, 3, 4) in P5 0.15 1.01 34.15 74.57
rational normal surface S(1, 4)⊂ P6

Q
1.41 0.74 5.32 0.06

Grassmannian G(1, 4)⊂ P9
Q

0.16 0.09 0.42 0.02
base locus of φ in Ex. 2.2 0.23 0.44 6.49 663.79
X ⊂ P11 in Ex. 2.3 over F3312 65.76 83.66 – 0.85
X ⊂ P11 in Ex. 2.3 over Z/16411 3.62 11.61 198.92 0.74

Table 2. Run-times to compute Segre classes in
CharacteristicClasses.m2, FMPIntersectionTheory.m2,
and Cremona.m2 (all times given in seconds).

https://github.com/Macaulay2/M2/blob/master/M2/Macaulay2/packages/CharacteristicClasses.m2
https://github.com/coreysharris/FMPIntersectionTheory-M2

SUPPLEMENT. The online supplement contains version 4.2.2 of Cremona.m2.

REFERENCES.
[Aluffi 2003] P. Aluffi, “Computing characteristic classes of projective schemes”, J. Symbolic Com-

put. 35:1 (2003), 3–19. MR Zbl

[Dolgachev 2011] I. Dolgachev, “Lectures on Cremona transformations”, 2011, available at http://
www.math.lsa.umich.edu/~idolga/cremonalect.pdf.

[Fulton 1984] W. Fulton, Intersection theory, Ergebnisse der Mathematik und ihrer Grenzgebiete (3)
2, Springer, 1984. MR Zbl

[Harris 1992] J. Harris, Algebraic geometry, a first course, Graduate Texts in Mathematics 133,
Springer, 1992. MR Zbl

[Harris 2017] C. Harris, “Computing Segre classes in arbitrary projective varieties”, J. Symbolic
Comput. 82 (2017), 26–37. MR Zbl

[Hartshorne 1977] R. Hartshorne, Algebraic geometry, Graduate Texts in Mathematics 52, Springer,
1977. MR Zbl

[Helmer 2016] M. Helmer, “Algorithms to compute the topological Euler characteristic, Chern–
Schwartz–MacPherson class and Segre class of projective varieties”, J. Symbolic Comput. 73 (2016),
120–138. MR Zbl

[Jost 2015] C. Jost, “Computing characteristic classes and the topological Euler characteristic of
complex projective schemes”, J. Softw. Algebra Geom. 7 (2015), 31–39. MR

[Macaulay2] D. R. Grayson and M. E. Stillman, “Macaulay2: a software system for research in
algebraic geometry”, available at http://www.math.uiuc.edu/Macaulay2.

[Miller and Sturmfels 2005] E. Miller and B. Sturmfels, Combinatorial commutative algebra, Grad-
uate Texts in Mathematics 227, Springer, 2005. MR Zbl

[Mumford 1988] D. Mumford, The red book of varieties and schemes, Lecture Notes in Mathematics
1358, Springer, 1988. MR Zbl

[Russo and Simis 2001] F. Russo and A. Simis, “On birational maps and Jacobian matrices”, Com-
positio Math. 126:3 (2001), 335–358. MR Zbl

[Russo and Staglianò 2017] F. Russo and G. Staglianò, “Congruences of 5-secant conics and the
rationality of some admissible cubic fourfolds”, 2017. arXiv

[Simis 2004] A. Simis, “Cremona transformations and some related algebras”, J. Algebra 280:1
(2004), 162–179. MR Zbl

[Staglianò 2016] G. Staglianò, “Examples of special quadratic birational transformations into com-
plete intersections of quadrics”, J. Symbolic Comput. 74 (2016), 635–649. MR Zbl

[Staglianò 2018] G. Staglianò, “Special cubic Cremona transformations of P6 and P7”, Adv. Geom.
(online publication March 2018).

RECEIVED: 19 Jan 2017 REVISED: 24 May 2018 ACCEPTED: 11 Jun 2018

GIOVANNI STAGLIANÒ:

giovannistagliano@gmail.com
Dipartimento di Ingegneria Industriale e Scienze Matematiche, Università Politecnica delle Marche,
Ancona, Italy

msp

http://msp.org/jsag/2018/8-1/jsag-v8-n1-x06-Cremona.zip
http://dx.doi.org/10.1016/S0747-7171(02)00089-5
http://msp.org/idx/mr/1956868
http://msp.org/idx/zbl/1074.14502
http://www.math.lsa.umich.edu/~idolga/cremonalect.pdf
http://dx.doi.org/10.1007/978-3-662-02421-8
http://msp.org/idx/mr/732620
http://msp.org/idx/zbl/0541.14005
http://dx.doi.org/10.1007/978-1-4757-2189-8
http://msp.org/idx/mr/1182558
http://msp.org/idx/zbl/0932.14001
http://dx.doi.org/10.1016/j.jsc.2016.09.003
http://msp.org/idx/mr/3608229
http://msp.org/idx/zbl/1357.14070
http://msp.org/idx/mr/0463157
http://msp.org/idx/zbl/0367.14001
http://dx.doi.org/10.1016/j.jsc.2015.03.007
http://dx.doi.org/10.1016/j.jsc.2015.03.007
http://msp.org/idx/mr/3385954
http://msp.org/idx/zbl/1349.14028
http://dx.doi.org/10.2140/jsag.2015.7.31
http://dx.doi.org/10.2140/jsag.2015.7.31
http://msp.org/idx/mr/3484270
http://www.math.uiuc.edu/Macaulay2
http://www.math.uiuc.edu/Macaulay2
http://msp.org/idx/mr/2110098
http://msp.org/idx/zbl/1090.13001
http://dx.doi.org/10.1007/978-3-662-21581-4
http://msp.org/idx/mr/971985
http://msp.org/idx/zbl/0658.14001
http://dx.doi.org/10.1023/A:1017572213947
http://msp.org/idx/mr/1834742
http://msp.org/idx/zbl/1036.14005
http://msp.org/idx/arx/1707.00999
http://dx.doi.org/10.1016/j.jalgebra.2004.03.025
http://msp.org/idx/mr/2081926
http://msp.org/idx/zbl/1067.14014
http://dx.doi.org/10.1016/j.jsc.2015.11.004
http://dx.doi.org/10.1016/j.jsc.2015.11.004
http://msp.org/idx/mr/3424062
http://msp.org/idx/zbl/1374.14014
http://dx.doi.org/10.1515/advgeom-2018-0001
mailto:giovannistagliano@gmail.com
http://msp.org


1





JOURNAL OF SOFTWARE FOR ALGEBRA AND GEOMETRY vol 8, no 1, 2018

1HeLP: a GAP package for torsion units in integral group rings
Andreas Bächle and Leo Margolis

11A software package to compute automorphisms of graded algebras
Simon Keicher

21A package for computations with classical resultants
Giovanni Staglianò

31The SpaceCurves package in Macaulay2
Mengyuan Zhang

49The ReesAlgebra package in Macaulay2
David Eisenbud

61A Macaulay2 package for computations with rational maps
Giovanni Staglianò

71ExteriorIdeals: a package for computing monomial ideals in an exterior algebra
Luca Amata and Marilena Crupi

81Software for computing conformal block divisors on M0,n
David Swinarski

87Divisor Package for Macaulay2
Karl Schwede and Zhaoning Yang

1948-7916(2018)8:1;1-F


	1. Description of the main methods
	From algebraic geometry to computational algebra
	Computing projective degrees
	Deterministic approach
	Probabilistic approach
	An alternative deterministic approach

	Some applications using projective degrees
	The degree of a rational map
	The Segre class

	Computing homogeneous components of kernels
	Inverting birational maps
	Heuristic approach


	2. Examples
	
	

